Rules for integrands of the form $\left(a + b x^n + c x^{2n}\right)^p$

1: $\int \left(a + b x^n + c x^{2n}\right)^p \, dx$ when $n < 0 \land p \in \mathbb{Z}$

Derivation: Algebraic expansion

Basis: If $p \in \mathbb{Z}$, then $\left(a + b x^n + c x^{2n}\right)^p == x^{2np} \left(c + b x^{-n} + a x^{-2n}\right)^p$

Rule 1.2.3.1.1: If $n < 0 \land p \in \mathbb{Z}$, then

$$\int \left(a + b x^n + c x^{2n}\right)^p dx \longrightarrow \int x^{2np} \left(c + b x^{-n} + a x^{-2n}\right)^p dx$$

Program code:

```
Int[(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  Int[x^(2*n*p)*(c+b*x^(-n)+a*x^(-2*n))^p,x] /;
FreeQ[{a,b,c},x] && EqQ[n2,2*n] && LtQ[n,0] && IntegerQ[p]
```

2: $\int \left(a + b x^n + c x^{2n}\right)^p \, dx$ when $n \in \mathbb{F}$

Derivation: Integration by substitution

Basis: If $k \in \mathbb{Z}^+$, then $F[x^n] == k \,\text{Subst}\left[x^{k-1} F\left[x^{kn}\right], x, x^{1/k}\right] \partial_x x^{1/k}$

Rule 1.2.3.1.2: If $b^2 - 4 a c \neq 0 \land n \in \mathbb{F}$, let $k = \text{Denominator}[n]$, then

$$\int \left(a + b x^n + c x^{2n}\right)^p dx \longrightarrow k \,\text{Subst}\left[\int x^{k-1} \left(a + b x^{kn} + c x^{2kn}\right)^p dx, x, x^{1/k}\right]$$

Program code:

```
Int[(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  With[{k=Denominator[n]},
  k*Subst[Int[x^(k-1)*(a+b*x^(k*n)+c*x^(2*k*n))^p,x],x,x^(1/k)]] /;
FreeQ[{a,b,c,p},x] && EqQ[n2,2*n] && FractionQ[n]
```

3: $\int (a + b x^n + c x^{2n})^p \, dx$ when $n \in \mathbb{Z}^-$

Derivation: Integration by substitution

Basis: If $n \in \mathbb{Z}$, then $F[x^n] = -\text{Subst}\left[\frac{F[x^{-n}]}{x^2}, x, \frac{1}{x}\right] \partial_x \frac{1}{x}$

Rule 1.2.3.1.3: If $n \in \mathbb{Z}^-$, then

$$\int (a + b x^n + c x^{2n})^p \, dx \longrightarrow -\text{Subst}\left[\int \frac{(a + b x^{-n} + c x^{-2n})^p}{x^2} \, dx, x, \frac{1}{x}\right]$$

Program code:

```
Int[(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  -Subst[Int[(a+b*x^(-n)+c*x^(-2*n))^p/x^2,x],x,1/x] /;
FreeQ[{a,b,c,p},x] && EqQ[n2,2*n] && ILtQ[n,0]
```

4: $\int \left(a + b\,x^n + c\,x^{2\,n}\right)^p\,dx$ when $b^2 - 4\,a\,c == 0$

Derivation: Piecewise constant extraction

Basis: If $b^2 - 4\,a\,c == 0$, then $\partial_x \frac{\left(a+b\,x^n+c\,x^{2\,n}\right)^p}{\left(b+2\,c\,x^n\right)^{2\,p}} == 0$

Note: If $b^2 - 4\,a\,c == 0$, then $a + b\,z + c\,z^2 == \frac{1}{4\,c}\,\left(b + 2\,c\,z\right)^2$

Rule 1.2.3.1.4: If $b^2 - 4\,a\,c == 0$, then

$$\int \left(a + b\,x^n + c\,x^{2\,n}\right)^p\,dx \longrightarrow \frac{\left(a + b\,x^n + c\,x^{2\,n}\right)^p}{\left(b + 2\,c\,x^n\right)^{2\,p}} \int \left(b + 2\,c\,x^n\right)^{2\,p}\,dx$$

Program code:

```
Int[(a_+b_.*x_^n_.+c_.*x_^n2_.)^p_,x_Symbol] :=
  (a+b*x^n+c*x^(2*n))^p/(b+2*c*x^n)^(2*p)*Int[(b+2*c*x^n)^(2*p),x] /;
FreeQ[{a,b,c,n,p},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0]
```

5. $\int (a + b x^n + c x^{2n})^p \, dx$ when $b^2 - 4 a c \neq 0 \land p \in \mathbb{Z}$

1: $\int (a + b x^n + c x^{2n})^p \, dx$ when $b^2 - 4 a c \neq 0 \land p \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.2.3.1.5.1: If $b^2 - 4 a c \neq 0 \land p \in \mathbb{Z}^+$, then

$$\int (a + b x^n + c x^{2n})^p \, dx \longrightarrow \int \text{ExpandIntegrand}\left[(a + b x^n + c x^{2n})^p, x\right] dx$$

Program code:

```
Int[(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x^n+c*x^(2*n))^p,x],x] /;
FreeQ[{a,b,c,n},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[p,0]
```

2: $\int \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\ \land\ p+1\in\mathbb{Z}^-$

Reference: G&R 2.161.5

Derivation: Trinomial recurrence 2b with $m=0$, $A=1$ and $B=0$

Note: G&R 2.161.4 is a special case of G&R 2.161.5.

Rule 1.2.3.1.5.2: If $b^2-4\,a\,c\neq 0\ \land\ p+1\in\mathbb{Z}^-$, then

$$\int \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx \longrightarrow$$
$$-\frac{x\,\left(b^2-2\,a\,c+b\,c\,x^n\right)\,\left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}}{a\,n\,(p+1)\,\left(b^2-4\,a\,c\right)}+$$
$$\frac{1}{a\,n\,(p+1)\,\left(b^2-4\,a\,c\right)}\int \left(b^2-2\,a\,c+n\,(p+1)\,\left(b^2-4\,a\,c\right)+b\,c\,(n\,(2\,p+3)+1)\,x^n\right)\,\left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}\,dx$$

Program code:

```
Int[(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  -x*(b^2-2*a*c+b*c*x^n)*(a+b*x^n+c*x^(2*n))^(p+1)/(a*n*(p+1)*(b^2-4*a*c)) +
  1/(a*n*(p+1)*(b^2-4*a*c))*
    Int[(b^2-2*a*c+n*(p+1)*(b^2-4*a*c)+b*c*(n*(2*p+3)+1)*x^n)*(a+b*x^n+c*x^(2*n))^(p+1),x] /;
FreeQ[{a,b,c,n},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && ILtQ[p,-1]
```

3. $\int \frac{1}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0$

1: $\int \frac{1}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\ \land\ \frac{n}{2}\in\mathbb{Z}^+\ \land\ b^2-4\,a\,c\not>0$

Derivation: Algebraic expansion

Basis: If $q\to\sqrt{\frac{a}{c}}$ and $r\to\sqrt{2\,q-\frac{b}{c}}$, then $\frac{1}{a+b\,z^2+c\,z^4}==\frac{r-z}{2\,c\,q\,r\,\left(q-r\,z+z^2\right)}+\frac{r+z}{2\,c\,q\,r\,\left(q+r\,z+z^2\right)}$

Note: If $(a \mid b \mid c) \in \mathbb{R} \land b^2 - 4 a c < 0$, then $\frac{a}{c} > 0$ and $2\sqrt{\frac{a}{c}} - \frac{b}{c} > 0$.

- Rule 1.2.3.1.5.3.1: If $b^2 - 4 a c \neq 0 \land \frac{n}{2} \in \mathbb{Z}^+ \land b^2 - 4 a c \not> 0$, let $q \to \sqrt{\frac{a}{c}}$ and $r \to \sqrt{2 q - \frac{b}{c}}$, then

$$\int \frac{1}{a + b x^n + c x^{2 n}} \, dx \longrightarrow \frac{1}{2 c q r} \int \frac{r - x^{n/2}}{q - r x^{n/2} + x^n} \, dx + \frac{1}{2 c q r} \int \frac{r + x^{n/2}}{q + r x^{n/2} + x^n} \, dx$$

- Program code:

```
Int[1/(a_+b_.*x_^n_+c_.*x_^n2_),x_Symbol] :=
  With[{q=Rt[a/c,2]},
  With[{r=Rt[2*q-b/c,2]},
  1/(2*c*q*r)*Int[(r-x^(n/2))/(q-r*x^(n/2)+x^n),x] +
  1/(2*c*q*r)*Int[(r+x^(n/2))/(q+r*x^(n/2)+x^n),x]]] /;
FreeQ[{a,b,c},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n/2,0] && NegQ[b^2-4*a*c]
```

2: $\int \frac{1}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ \left(\frac{n}{2}\notin\mathbb{Z}^+\vee\ b^2-4\,a\,c>0\right)$

Reference: G&R 2.161.1a

Derivation: Algebraic expansion

Basis: Let $q\to\sqrt{b^2-4\,a\,c}$, then $\frac{1}{a+b\,z+c\,z^2}==\frac{c}{q}\,\frac{1}{\frac{b}{2}-\frac{q}{2}+c\,z}-\frac{c}{q}\,\frac{1}{\frac{b}{2}+\frac{q}{2}+c\,z}$

Rule 1.2.3.1.5.3.2: If $b^2-4\,a\,c\neq 0$, let $q\to\sqrt{b^2-4\,a\,c}$, then

$$\int\frac{1}{a+b\,x^n+c\,x^{2\,n}}\,dx\ \longrightarrow\ \frac{c}{q}\int\frac{1}{\frac{b}{2}-\frac{q}{2}+c\,x^n}\,dx-\frac{c}{q}\int\frac{1}{\frac{b}{2}+\frac{q}{2}+c\,x^n}\,dx$$

Program code:

```
Int[1/(a_+b_.*x_^n_+c_.*x_^n2_),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  c/q*Int[1/(b/2-q/2+c*x^n),x] - c/q*Int[1/(b/2+q/2+c*x^n),x]] /;
FreeQ[{a,b,c},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0]
```

6: $\int\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x\frac{\left(a+b\,x^n+c\,x^{2\,n}\right)^p}{\left(1+\frac{2\,c\,x^n}{b+\sqrt{b^2-4\,a\,c}}\right)^p\left(1+\frac{2\,c\,x^n}{b-\sqrt{b^2-4\,a\,c}}\right)^p}==0$

Rule 1.2.3.1.6: If $b^2-4\,a\,c\neq 0\ \wedge\ p\notin\mathbb{Z}$, then

$$\int \left(a + b x^n + c x^{2n}\right)^p dx \longrightarrow \frac{a^{\text{IntPart}[p]} \left(a + b x^n + c x^{2n}\right)^{\text{FracPart}[p]}}{\left(1 + \frac{2 c x^n}{b+\sqrt{b^2-4 a c}}\right)^{\text{FracPart}[p]} \left(1 + \frac{2 c x^n}{b-\sqrt{b^2-4 a c}}\right)^{\text{FracPart}[p]}} \int \left(1 + \frac{2 c x^n}{b + \sqrt{b^2 - 4 a c}}\right)^p \left(1 + \frac{2 c x^n}{b - \sqrt{b^2 - 4 a c}}\right)^p dx$$

Program code:

```
Int[(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  a^IntPart[p]*(a+b*x^n+c*x^(2*n))^FracPart[p]/
    ((1+2*c*x^n/(b+Rt[b^2-4*a*c,2]))^FracPart[p]*(1+2*c*x^n/(b-Rt[b^2-4*a*c,2]))^FracPart[p])*
    Int[(1+2*c*x^n/(b+Sqrt[b^2-4*a*c]))^p*(1+2*c*x^n/(b-Sqrt[b^2-4*a*c]))^p,x] /;
FreeQ[{a,b,c,n,p},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

S: $\int \left(a + b u^n + c u^{2n}\right)^p dx$ when $u == d + e x$

Derivation: Integration by substitution

Rule 1.2.3.1.S: If $u == d + e x$, then

$$\int \left(a + b u^n + c u^{2n}\right)^p dx \longrightarrow \frac{1}{e} \text{Subst}\left[\int \left(a + b x^n + c x^{2n}\right)^p dx,\ x,\ u\right]$$

Program code:

```
Int[(a_+b_.*u_^n_+c_.*u_^n2_.)^p_,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(a+b*x^n+c*x^(2*n))^p,x],x,u] /;
FreeQ[{a,b,c,n,p},x] && EqQ[n2,2*n] && LinearQ[u,x] && NeQ[u,x]
```

9. $\int (a + b\,x^{-n} + c\,x^{n})^{p}\,dx$

1: $\int \left(a + b\,x^{-n} + c\,x^{n}\right)^{p}\,dx$ when $p \in \mathbb{Z}$

Derivation: Algebraic normalization

Basis: $a + b\,x^{-n} + c\,x^{n} = \frac{b + a\,x^{n} + c\,x^{2\,n}}{x^{n}}$

Rule 1.2.3.1.9.1: If $p \in \mathbb{Z}$, then

$$\int \left(a + b\,x^{-n} + c\,x^{n}\right)^{p}\,dx \longrightarrow \int \frac{\left(b + a\,x^{n} + c\,x^{2\,n}\right)^{p}}{x^{n\,p}}\,dx$$

–

Program code:

```
Int[(a_+b_.*x_^mn_+c_.*x_^n_.)^p_.,x_Symbol] :=
  Int[(b+a*x^n+c*x^(2*n))^p/x^(n*p),x] /;
FreeQ[{a,b,c,n},x] && EqQ[mn,-n] && IntegerQ[p] && PosQ[n]
```

2: $\int \left(a + b\,x^{-n} + c\,x^{n}\right)^{p}\,dx$ when $p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{x^{n\,p}\,\left(a+b\,x^{-n}+c\,x^{n}\right)^{p}}{\left(b+a\,x^{n}+c\,x^{2\,n}\right)^{p}} == 0$

Basis: $\frac{x^{n\,p}\,\left(a+b\,x^{-n}+c\,x^{n}\right)^{p}}{\left(b+a\,x^{n}+c\,x^{2\,n}\right)^{p}} == \frac{x^{n\,\text{FracPart}[p]}\,\left(a+b\,x^{-n}+c\,x^{n}\right)^{\text{FracPart}[p]}}{\left(b+a\,x^{n}+c\,x^{2\,n}\right)^{\text{FracPart}[p]}}$

Rule 1.2.3.1.9.2: If $p \notin \mathbb{Z}$, then

$$\int \left(a + b\,x^{-n} + c\,x^{n}\right)^{p}\,dx \longrightarrow \frac{x^{n\,\text{FracPart}[p]}\,\left(a + b\,x^{-n} + c\,x^{n}\right)^{\text{FracPart}[p]}}{\left(b + a\,x^{n} + c\,x^{2\,n}\right)^{\text{FracPart}[p]}} \int \frac{\left(b + a\,x^{n} + c\,x^{2\,n}\right)^{p}}{x^{n\,p}}\,dx$$

–

Program code:

```
Int[(a_+b_.*x_^mn_+c_.*x_^n_.)^p_,x_Symbol] :=
  x^(n*FracPart[p])*(a+b*x^(-n)+c*x^n)^FracPart[p]/(b+a*x^n+c*x^(2*n))^FracPart[p]*Int[(b+a*x^n+c*x^(2*n))^p/x^(n*p),x] /;
FreeQ[{a,b,c,n,p},x] && EqQ[mn,-n] && Not[IntegerQ[p]] && PosQ[n]
```